
ABSTRACT: In this Macaulay2 package we implement a type of object called a LinearCode. We
implement functions that compute basic parameters and objects associated with a linear code, such as
generator and parity check matrices, the dual code, length, dimension, and minimum distance, among
others. We implement a type of object called an EvaluationCode, a construction which allows users to
study linear codes using tools of algebraic geometry and commutative algebra. We implement functions
to generate important families of linear codes, such as Hamming codes, cyclic codes, Reed–Solomon
codes, Reed–Muller codes, Cartesian codes, monomial–Cartesian codes, and toric codes. In addition,
we implement functions for the syndrome decoding algorithm and locally recoverable code construction,
which are important tools in applications of linear codes.

1. INTRODUCTION. Coding theory has been extensively studied since 1948, when Claude Shannon
[1948] proved in his seminal paper that linear codes can be used to reliably transmit information from
a single source to a single receiver through a noisy channel. Since then, coding theory has found many
important engineering applications. For example, coding theory has been used in designing reliable
data storage systems, radio communication protocols, and in the emerging field of quantum computers.
Coding theory has close ties with many areas in mathematics including linear algebra, commutative
algebra, algebraic geometry, and combinatorics.

In this note we introduce the new [Macaulay2] package called CodingTheory. The goal of this pack-
age is to provide a range of functions for constructing linear and evaluation codes over finite fields, and for
computing some of their main properties. To this aim, we implement two types of objects, LinearCode
and EvaluationCode. The package also includes implementations of functions for generating important
families of linear codes like Hamming codes, cyclic codes, Reed–Solomon codes, Reed–Muller codes,
Cartesian codes, monomial-Cartesian codes and toric codes. It also has functions for the syndrome
decoding algorithm and locally recoverable codes.

The organization of this note is as follows. In Section 2 we describe various ways to construct a linear
code over a finite field using the CodingTheory package. In Section 3 we show how to compute the
main parameters of a linear code: length, dimension, and minimum distance. We also illustrate how to

compute some of the main algebraic objects associated with linear codes, such as generator and parity
check matrices, dual codes, etc. In Section 4 we give a brief introduction to evaluation codes and describe
some functions implemented to study these objects. In Section 5 we explain how to create some of the
most studied families of linear codes, including Hamming codes, cyclic codes, Reed–Solomon codes,
and Reed–Muller codes. Finally, we give instructions on how to create locally recoverable codes.

In this paper we do not attempt to fully explain every function distributed in this package. For a
detailed explanation of all functions in the package, we refer to the Macaulay2 help page which can be
accessed by running

i1: viewHelp CodingTheory

More information about basics of coding theory can be found in [Huffman and Pless 2003; MacWilliams
and Sloane 1977; van Lint 1999]. Constructions of codes using commutative algebra as evaluation codes
can be seen in [Carvalho et al. 2017; Gold et al. 2005; Hansen 2000; Little and Schenck 2006; Martínez-
Bernal et al. 2017; 2018; Rentería-Márquez et al. 2011; Rentería and Tapia-Recillas 1997; Ruano 2007;
Soprunov and Soprunova 2009; Soprunov 2013]. Excellent references for the theory of vanishing ideals
and their properties are [Cox et al. 1992; Villarreal 2015].

2. CONSTRUCTING LINEAR CODES. Let Fq be a finite field with q elements. Mathematically, a linear
code is defined as a vector subspace C ⊆ Fn

q and it is often specified by a generator matrix, which is a k×n
matrix G with entries in Fq whose k rows form a basis for C . In Macaulay2, a linear code is defined
as an Fq-submodule of Fn

q using the constructor linearCode. This constructor is an instance of the
LinearCode type. There are various ways to use the command linearCode. For example, one can use
this command to construct a linear code C ⊆ Fn

q by specifying a generator matrix G of C (Example 2.1).
Alternatively, one can use the command linearCode to construct a linear code C ⊆ Fn

q by indicating
the finite field Fq and a list L of elements of Fn

q that span C (Example 2.2). More details and equivalent
ways to use the constructor linearCode are given next.

Inputs: Usage:

• F = GF(q), a finite field with q elements • linearCode(G)

• n, r, p, positive integers with p prime • linearCode(F,L)

• G, a matrix with entries in GF(q) • linearCode(F,n,L)

• L, a list of elements of GF(q)n
• linearCode(p,r,n,L)

In the next examples we construct a simple binary linear code C ⊆ F4
2 with generator matrix G =(

1
0

1
0

0
1

0
1

)
using equivalent versions of the command linearCode.

Example 2.1.
i1 : F = GF(2);

i2 : L = {{1,1,0,0},{0,0,1,1}};

i3 : G = matrix apply(L,codeword->apply(codeword,entry->sub(entry,F)));

i4 : C = linearCode(G);

i5 : C.GeneratorMatrix
o5 = | 1 1 0 0 |

| 0 0 1 1 |

Note that in Example 2.1 it was necessary to coerce the entries of each vector in the list L into elements
of F = GF(2). An equivalent way to do this is to pass the field GF(q) to the matrix constructor or to
use the constructor linearCode(GF(q),L).

Example 2.2.
i1 : L = {{1,1,0,0},{0,0,1,1}};

i2 : C = linearCode(GF(2),L);

i3 : C.GeneratorMatrix
o3 = | 1 1 0 0 |

| 0 0 1 1 |

The set F∗q of nonzero elements of a finite field Fq is a multiplicative cyclic group ([Huffman and
Pless 2003, Theorem 3.3.1]). A generator of F∗q is called a primitive element of Fq . One way to refer
to a primitive element of a finite field is by specifying a symbol using the Variable option of the
constructor GF. In the next example we illustrate how to define a linear code C ⊆ F10

11 with generator
matrix

G =


1 a1 a2

· · · a9

1 a2 a4
· · · (a2)9

1 a3 (a3)2
· · · (a3)9

1 a4 (a4)2
· · · (a4)9

,

where a is a primitive element of F11. In Macaulay2, a = 2.

Example 2.3.
i1 : F = GF(11,Variable => a);

i2 : G = matrix table({1,2,3,4},
{1,a,a^2,a^3,a^4,a^5,a^6,a^7,a^8,a^9},(i,j)->j^i);

i3 : C = linearCode(G);

i4 : C.GeneratorMatrix
o4 = | 1 2 4 -3 5 -1 -2 -4 3 -5 |

| 1 4 5 -2 3 1 4 5 -2 3 |
| 1 -3 -2 -5 4 -1 3 2 5 -4 |
| 1 5 3 4 -2 1 5 3 4 -2 |

In Fn
q there is a standard inner product defined for all x = (x1, . . . , xn) and y = (y1, . . . , yn) in Fn

q as
x · y = x1 y1+ · · ·+ xn yn . Given a linear code C ⊆ Fn

q , the dual or orthogonal code of C with respect to
this inner product is the linear code C⊥ defined as

C⊥ = {x ∈ Fn
q : x · c = 0 for all c ∈ C}.

A generator matrix of C⊥ is called a parity check matrix for C . Since (C⊥)⊥ = C , another common way
to mathematically define C is by specifying one of its parity check matrices. By using the command
linearCode and the ParityCheck option of this command, we can define a linear code C ⊆ Fn

q by
either specifying a parity check matrix H of C or a list L of elements of Fn

q that span C⊥.

In the next example, we define a linear code C ⊆ F5
9 whose dual code C⊥ is generated by the set

{(1, 0, a, 0, 0), (0, 1, a+1, 1, 0), (1, 1, 1, a, 0)}⊆ F5
9, where a is a primitive element of F9. In Macaulay2,

a ∈ F9 satisfies a2
= a+ 1.

Example 2.4.
i1 : F = GF(9,Variable => a);

i2 : L = {{1,0,a,0,0},{0,a,a+1,1,0},{1,1,1,a,0}};

i3 : C = linearCode(F,L,ParityCheck => true);

i4 : C.GeneratorMatrix
o4 = | a-1 0 a+1 1 0 |

| 0 0 0 0 1 |

i5 : C.ParityCheckMatrix
o5 = | 1 0 a 0 0 |

| 0 a a+1 1 0 |
| 1 1 1 a 0 |

Although the dual code of a linear code can be constructed using the command dualCode implemented
in the CodingTheory package, the ParityCheck option of the command linearCode also allows us
to construct the dual of a linear code. In the next example we construct the dual of the linear code in
Example 2.3.

Example 2.5.
i1 : F = GF(11,Variable => a);

i2 : G = matrix table({1,2,3,4},
{1,a,a^2,a^3,a^4,a^5,a^6,a^7,a^8,a^9},(i,j)->j^i);

i3 : D = linearCode(G,ParityCheck => true);

i4 : D.GeneratorMatrix
o4 = | 1 -3 5 3 1 0 0 0 0 0 |

| -3 -1 4 -4 0 1 0 0 0 0 |
| 4 -4 -3 5 0 0 1 0 0 0 |
| -5 -3 4 4 0 0 0 1 0 0 |
| -4 -4 -1 3 0 0 0 0 1 0 |
| -3 5 3 1 0 0 0 0 0 1 |

3. BASIC PARAMETERS OF LINEAR CODES. The dimension k and length n are basic parameters of a
linear code C ⊆ Fn

q . The information rate of a linear code is defined as k/n. Another parameter of a
linear code C ⊆ Fn

q is the minimum distance, which is defined as

wH (C) :=min{‖c‖ : c ∈ C, c 6= 0},

where ‖c‖ denotes the Hamming weight of c ∈ Fn
q , that is, ‖c‖ is the number of nonzero entries in c.

This parameter is important in determining the error-correcting capability of C ; the higher the minimum
distance, the more errors the code can detect and correct (see [Huffman and Pless 2003]). While the
dimension and the length of linear code are computationally easy to determine, it is known that computing
the minimum distance of an arbitrary linear code is an NP-hard problem [Vardy 1997]. For this task, the
function minimumWeight is provided.

In Macaulay2, the space Fn
q has been called the ambient module or ambient space of C . The field Fq

is called the alphabet or field of C . The elements of C are referred to as codewords. The input in all the

following commands implemented in the CodingTheory package is always a linear code C:

• dim C • field C • C.AmbientModule

• length C • codewords C • minimumWeight C

• alphabet C • C.Generators • informationRate C

• ambientSpace C • C.GeneratorMatrix • C.ParityCheckMatrix

Example 3.1.

i1 : L = {{1,1,0,0},{0,0,1,1}};

i2 : C = linearCode(GF(4),L);

i3 : dim C
o3 : 2

i4 : length C
o4 = 4

i5 : alphabet C
o5 = {0, a, a + 1, 1}

i6 : ambientSpace C
4

o6 = (GF 4)

i7 : field C
o7 = GF 4

i8 : minimumWeight C
o8 = 2

i9 : codewords C
o9 = {{1, 1, 1, 1}, {1, 1, a, a}, {a, a, 1, 1}, {a, a, a, a},

------------------------------------------------------------------
{a + 1, a + 1, a, a}, {a + 1, a + 1, 1, 1}, {1, 1, a + 1, a + 1},

------------------------------------------------------------------
{a, a, a + 1, a + 1}, {a + 1, a + 1, a + 1, a + 1}, {1, 1, 0, 0},

------------------------------------------------------------------
{0, 0, 1, 1}, {0, 0, a, a}, {a, a, 0, 0}, {a + 1, a + 1, 0, 0},

------------------------------------------------------------------
{0, 0, a + 1, a + 1}, {0, 0, 0, 0}}

4. EVALUATION CODES. Let P = {a1, . . . , an} be a subset of Fm
q . Consider a finite-dimensional sub-

space S ⊂ Fq [X1, . . . , Xm] of the ring of polynomials over Fq in m variables. The evaluation map

evS : S→ Fq
|P|, f 7→ ( f (a1), . . . , f (an)),

defines a linear map of Fq -vector spaces. The image of evS in Fq
|P|, denoted by CP(S), is the evaluation

code on the set P corresponding to S. The vanishing ideal of P, denoted by I (P), is the ideal in
Fq [X1, . . . , Xn] of all polynomials that vanish on P. A key observation that allows the use of commutative
algebra in studying evaluation codes is that the kernel of the evaluation map evS is precisely S ∩ I (P).

An evaluation code CP(S) is implemented in Macaulay2 as an object of type EvaluationCode.
However, the object C.LinearCode is a linear code in Macaulay2. This has been done in this way
because there are more objects associated with an evaluation code than with a linear code. For instance,
the vanishing ideal associated to the set P plays an important role when finding and estimating parameters
of the code, so it is convenient to be able to access it.

There are many constructions of evaluation codes for specific choices of the set P and the subspace S.
These include Reed–Muller codes, Cartesian, monomial Cartesian codes, toric codes, and evaluation
codes from graphs. We refer to [Carvalho et al. 2017; Hansen 2000; Little and Schenck 2006; López
et al. 2014; Martínez-Bernal et al. 2017; Rentería-Márquez et al. 2011; Ruano 2007; Soprunov and
Soprunova 2009] for details on how these codes are defined and what properties they have from coding
theory, commutative algebra, and algebraic geometry perspectives.

Some functions implemented in the CodingTheory package for various constructions of evaluation
codes and associated algebraic objects are the following:

Inputs:

• I, an ideal • M, an integer matrix

• d,r, positive integers • L, a list of subsets of F

• F=GF(q), a finite field with q elements • v, a list of m positive integers

• P, a list of points in Fm
• MI, an incident matrix of a graph

• S, a list of polynomials in m variables

Usage:

• evaluationCode(F,P,S) • vNumber(I)

• toricCode(F,M) • footPrint(d,r,I)

• cartesianCode(F,L,d) • hYp(d,r,I)

• orderCode(F,P,v,d) • genMinDisIdeal(d,r,I)

• evCodeGraph(F,MI,S) • vasconcelosDegree(d,r,I)

The input S above is a list of polynomials that span the subspace

S ⊂ Fq [X1, . . . , Xm].

The mathematical definitions of the functions in the second column of the previous list can be found in
[Cooper et al. 2020]. The following example shows how to construct an evaluation code in Macaulay2
using the CodingTheory package.

Example 4.1.
i1 : F=GF(4,Variable=>a); R=F[x,y,z];

i3 : P={{0,0,0},{1,0,0},{0,1,0},{0,0,1},{1,1,1},{a,a,a}};

i4 : S={x+y+z,a+y*z^2,z^2,x+y+z+z^2};

i5 : C=evaluationCode(F,P,S);

i6 : (C.LinearCode).GeneratorMatrix
o6 = | 0 1 1 1 1 a |

| a a a a a+1 a+1 |
| 0 0 0 1 1 a+1 |
| 0 1 1 0 0 1 |

i7 : length C.LinearCode
o7 = 6

i8 : dim C.LinearCode
o8 = 3

i9 : C.Points
o9 = {{0, 0, 0}, {1, 0, 0}, {0, 1, 0}, {0, 0, 1}, {1, 1, 1}, {a, a, a}}

i10 : C.VanishingIdeal
2 2 2 2

o10 = ideal (x*z + y*z, y + z + y + z, x*y + y*z, x + z + x + z,
--------------------------------------------------------------
3 2

z + (a +1)z + a*z)

5. FAMILIES OF LINEAR CODES. Various important families of linear codes have been defined through
the development of coding theory. These include Hamming codes, cyclic codes, Reed–Solomon codes,
Reed–Muller codes, etc. The mathematical definitions of these and many other families of codes can be
found in [Huffman and Pless 2003; MacWilliams and Sloane 1977; van Lint 1999].

The following functions have been implemented in the CodingTheory package to construct some of
these important families of codes.

Inputs:

• n, k, d, m, s, positive integers • F = GF(q), a finite field with q elements

• g(x), a polynomial in F[x] • E, a list of elements of F

• L, a list of vectors in Fm

Usage:

• hammingCode(q,s) • cyclicCode(F,g(x),n) • zeroCode(F,n)

• reedSolomonCode(F,E,s) • repetitionCode(F,n) • universeCode(F,n)

• reedMullerCode(q,m,d) • randomCode(F,n,k) • zeroSumCode(F,n)

Example 5.1.
i1 : C = hammingCode(2,3);

i2 : C.GeneratorMatrix
o2 = | 1 1 1 1 0 0 0 |

| 1 1 0 0 1 0 0 |
| 0 1 1 0 0 1 0 |
| 1 0 1 0 0 0 1 |

i3 : F = GF(5); R = F[x]; g = x-1; C = cyclicCode(F,g,6);

i7 : C.GeneratorMatrix
o7 = | -1 1 0 0 0 0 |

| 0 -1 1 0 0 0 |
| 0 0 -1 1 0 0 |
| 0 0 0 -1 1 0 |
| 0 0 0 0 -1 1 |

i8 : C = reedSolomonCode(GF(5),{1,2,3},3);

i9 : (C.LinearCode).GeneratorMatrix
o9 = | 1 1 1 |

| 1 2 -2 |
| 1 -1 -1 |

6. APPLICATIONS OF LINEAR CODES. An important aspect in coding theory is decoding, which is
used when information is transmitted trough a noisy channel. In a few words the idea is the following.
Take a vector c ∈ C . Change the value of some of the entries of c to obtain a new vector v. Decoding
the vector v means to recover the vector c when only v and C are given. Detailed treatment of decoding
algorithms can be found in [Huffman and Pless 2003]. Another, more recent application of coding theory
is found in distributed and cloud storage systems. The idea is to use locally recoverable codes, which
are linear codes with the property that every entry can be recovered from a few other entries. For more
information on locally recoverable codes, see [Tamo and Barg 2014].

Some of the most important functions implemented in the CodingTheory package that can be used for
applications of coding theory are the following:

Inputs:

• C, a linear code over GF(q)

• v, a vector in the ambient space of C

• {q,n,k,r}, where q is a prime power, and n, k, and r are positive integers

• L, a list of pairwise disjoint subsets of GF(q)

Usage:

• syndromeDecode(C,v,minimumWeight(C))

• LocallyRecoverableCode({q,n,k,r},L,a polynomial)

Example 6.1.

i1 : C = hammingCode(2,3);

i2 : msg = matrix {{1,0,1,0}};

i3 : v = msg*(C.GeneratorMatrix)
o3 = | 0 1 0 1 0 1 0 |

i4 : err = matrix take(random entries basis source v, 1)
o4 = | 0 0 0 0 1 0 0 |

i5 : received = transpose(transpose (v+err))
o5 = | 0 1 0 1 1 1 0 |

i6 : transpose syndromeDecode(C, transpose recieved, 3)
o6 = | 0 1 0 1 0 1 0 |

